Cálculo Diferencial e Integral: um KIT de sobrevivência

This woksheet is in Portuguese language.

Prof. Doherty Andrade

Transformada de Laplace

Nesta woksheet está todo o material visto em aula sobre transformada de Laplace.

Use-o como uma revisão, mas não esqueça de lapis e papel.

Transformada de Laplace

Teorema (Existencia da Transformada de Laplace) Se $f(t)$ é de ordem exponencial,

então sua transformada de Laplace $Lf(t) = F(s)$ é dada por

$$F(s) = \int_0^{\infty} f(t)\, e^{(-s\,t)}\, dt \quad .$$

A integral definindo $F(s)$ existe nos pontos $\tau < s$.

Teorema (Linearidade da Transformada de Laplace) Sejam $f(t)$ e $g(t)$ tendo transformada de Laplace dadas por $F(s)$ e $G(s)$, respectivamente. Se a e b são constantes, então $L(a\,f(t) + b\,g(t)) = a\,F(s) + b\,G(s)$.

Teorema (Unicidade da Transformada de Laplace) Sejam $f(t)$ e $g(t)$ tendo Transformada de Laplace dadas por $F(s)$ e $G(s)$, respectivamente. Se $F(s) = G(s)$ então $f(t) = g(t)$ **.**

Para trabalhar com Transformada de Laplace no Maple, você precisa carregar os procedimentos "Laplace transform" .

Faça isto com

```
> with(inttrans):
```

Exemplo 1 Determine a transformada de Laplace da função degrau unitário.

$f(t) = 1$ se $0 \leq t < c$,

$f(t) = 0$ se $c < t$.

```
> c:='c': f:='f': F:='F': g:='g': s:='s': t:='t': T:='T':
  f0 := t -> 1:
  g := t -> subs(T=t, int(f0(T)*exp(-s*T),T)):
  F := t -> subs(T=t, int(f0(T)*exp(-s*T),T=0..c)):
  `For 0 <= t <= c,  f(t) ` = f0(t);
  Int(f(t)*exp(-s*t),t) = g(t);
  `F(s) = `, Int(f(t)*exp(-s*t),t=0..c) = F(s);
  `F(s) ` = simplify(F(s));
```

$$For\ 0 <= t <= c,\ f(t) = 1$$

$$\int f(t)\, e^{(-s\, t)}\, dt = -\frac{e^{(-s\, t)}}{s}$$

$$F(s) = ,\ \int_0^c f(t)\, e^{(-s\, t)}\, dt = -\frac{e^{(-c\, s)}}{s} + \frac{1}{s}$$

$$F(s) = -\frac{e^{(-c\, s)} - 1}{s}$$

Veja o gráfico de f.

```
> f:=x -> piecewise(x>0,1);#tomei c=0 aqui
```

$$f := x \to \mathrm{piecewise}(0 < x,\ 1)$$

```
> with(plots):
```

```
> plot(f(x),x=-2..4,y=0..2);
```



Exemplo 2 Determine a transformada de Laplace de $\mathrm{f}(t) = e^{(a\,t)}$.

```
> a:='a': f:='f': F:='F': g:='g': s:='s': t:='t': T:='T':
  f0 := t -> exp(a*t):
  `f(t) ` = f0(t);
  g := proc(t,S)
    simplify(subs(T=t,int(f0(T)*exp(-S*T),T)))
  end:
  Int(f(t)*exp(-s*t),t) = g(t,s);
  `F(s) ` = subs(T=t, int(f0(T)*exp(-s*T),T=0..infinity));
  `F(s) ` = simplify(g(infinity,s) - g(0,s));
  F := s -> - subs(S=s, g(0,S)):
  `F(s) ` = F(s);
```

$$f(t) = \mathbf{e}^{(a\,t)}$$

$$\int \mathrm{f}(t)\,\mathbf{e}^{(-s\,t)}\,dt = \frac{\mathbf{e}^{(t\,(a-s))}}{a-s}$$

$$F(s) = \lim_{t \to \infty} \frac{\mathbf{e}^{(a\,t - s\,t)}}{a-s} - \frac{1}{a-s}$$

$$F(s) = \frac{e^{(\infty\,(a-s))} - 1}{a-s}$$

$$F(s) = -\frac{1}{a-s}$$

```
> `f(t) ` = exp(a*t);
  `F(s) ` = laplace(exp(a*t), t, s);
```

$$f(t) = e^{(a\,t)}$$

$$F(s) = \frac{1}{-a+s}$$

Exemplo 3 Determine a transformada de Laplace de $f(t) = \sinh(a\,t)$.

Como $f(t) = \sinh(at) = \frac{e^{(a\,t)} - e^{(-a\,t)}}{2}$, usamos que

$L_1(e^{(a\,t)}) = \frac{1}{s-a}$ e $L_2(e^{(-a\,t)}) = \frac{1}{s+a}$.

```
> L:='L':
  `f(t) ` = sinh(a*t);
  `f(t) ` = (exp(a*t)-exp(-a*t))/2;
  L1 :=laplace( exp(a*t), t, s):
  L2 :=laplace(exp(-a*t), t, s):
  L(exp(a*t)) = L1;
  L(exp(-a*t)) = L2; ` `;
  `F(s) ` = (L(exp(a*t)) - L(exp(-a*t)))/2;
  `F(s) ` = (L1 - L2)/2;
  `F(s) ` = simplify((L1 - L2)/2);
```

$$f(t) = \sinh(a\,t)$$

$$f(t) = \frac{1}{2}e^{(a\,t)} - \frac{1}{2}e^{(-a\,t)}$$

$$L(e^{(a\,t)}) = \frac{1}{-a+s}$$

$$\mathrm{L}(e^{(-a\,t)}) = \frac{1}{s+a}$$

$$F(s) = \frac{1}{2}\mathrm{L}(e^{(a\,t)}) - \frac{1}{2}\mathrm{L}(e^{(-a\,t)})$$

$$F(s) = \frac{1}{2}\frac{1}{-a+s} - \frac{1}{2}\frac{1}{s+a}$$

$$F(s) = -\frac{a}{(a-s)\,(s+a)}$$

Podemos verificar este resultado usando as rotinas do Maple.

```
> `f(t) ` = sinh(a*t);
  `F(s) ` = laplace(sinh(a*t), t, s);
```

$$f(t) = \sinh(a\,t)$$

$$F(s) = \frac{a}{s^2 - a^2}$$

Exemplo 4 Determine a trasnformada de Laplace de $f(t) = t$.

```
> a:='a': f:='f': F:='F': g:='g': s:='s': t:='t': T:='T':
  f0 := t -> t:
  `f(t) ` = f0(t);
  g := proc(t,S)
    simplify(subs(T=t,int(f0(T)*exp(-S*T),T)))
  end:
  Int(f(t)*exp(-s*t),t) = g(t,s);
  `F(s) ` = subs(T=t, int(f0(T)*exp(-s*T),T=0..infinity));
  `F(s) ` = simplify(g(infinity,s) - g(0,s));
  F := s -> - subs(S=s, g(0,S)):
  `F(s) ` = F(s);
```

$$f(t) = t$$

$$\int f(t)\, e^{(-s\,t)}\, dt = -\frac{e^{(-s\,t)}\,(s\,t+1)}{s^2}$$

$$F(s) = \lim_{t \to \infty} \frac{-s\,t\,e^{(-s\,t)} - e^{(-s\,t)}}{s^2} + \frac{1}{s^2}$$

$$F(s) = -\frac{s\,\infty\,e^{(-s\,\infty)} + e^{(-s\,\infty)} - 1}{s^2}$$

$$F(s) = \frac{1}{s^2}$$

Podemos verificar este resultado usando as rotinas do pacote Transformada de Laplace.

```
> `f(t) ` = t;
  `F(s) ` = laplace(t, t, s);
```

$$f(t) = t$$

$$F(s) = \frac{1}{s^2}$$

Exemplo 5 Determine a transformada de Laplace de $f(t) = \cos(b\,t)$.

```
> a:='a': f:='f': F:='F': g:='g': s:='s': t:='t': T:='T':
  f0 := t -> cos(b*t):
  `f(t) ` = f0(t);
  g := proc(t,S)
    simplify(subs(T=t,int(f0(T)*exp(-S*T),T)))
  end:
  Int(f(t)*exp(-s*t),t) = g(t,s);
  `F(s) ` = subs(T=t, int(f0(T)*exp(-s*T),T=0..infinity));
  `F(s) ` = simplify(g(infinity,s) - g(0,s));
  F := s -> - subs(S=s, g(0,S)):
  `F(s) ` = F(s);
```

$$f(t) = \cos(b\,t)$$

$$\int f(t)\,e^{(-s\,t)}\,dt = \frac{e^{(-s\,t)}\,(-s\cos(b\,t) + b\sin(b\,t))}{s^2 + b^2}$$

$$F(s) = \lim_{t \to \infty} -\frac{s\,e^{(-s\,t)}\cos(b\,t)}{s^2 + b^2} + \frac{b\,e^{(-s\,t)}\sin(b\,t)}{s^2 + b^2} + \frac{s}{s^2 + b^2}$$

$$F(s) = \frac{-s\, e^{(-s\,\infty)} \cos(b\,\infty) + b\, e^{(-s\,\infty)} \sin(b\,\infty) + s}{s^2 + b^2}$$

$$F(s) = \frac{s}{s^2 + b^2}$$

Verfique este resultado usando as rotinas do Maple.

```
> `f(t) ` = cos(b*t);
  `F(s) ` = laplace(cos(b*t), t, s);
```

$$f(t) = \cos(b\,t)$$

$$F(s) = \frac{s}{s^2 + b^2}$$

Exemplo 6 Determine a transformada inversa de $\mathrm{F}(s) = \dfrac{3\,s + 6}{s^2 + 9}$.

```
> f:='f': F:='F': s:='s': t:='t':
  F0 := s -> (3*s + 6)/(s^2 + 9):
  `F(s) ` = F0(s);
  `F(s) ` = expand(F0(s));
```

$$F(s) = \frac{3\,s + 6}{s^2 + 9}$$

$$F(s) = 3\,\frac{s}{s^2 + 9} + \frac{6}{s^2 + 9}$$

A transformada $\mathrm{F}(s)$ é uma combinação linear .

```
> F1 := s/(s^2 + 9):
  F2 := 3/(s^2 + 9):
  F[1](s) = F1;
  F[2](s) = F2;
  `F(s) = `, 3*F[1](s) + 2*F[2](s) = 3*F1 + 2*F2;
```

$$F_1(s) = \frac{s}{s^2 + 9}$$

$$F_2(s) = \frac{3}{s^2+9}$$

$$F(s) = ,\ 3\,F_1(s) + 2\,F_2(s) = 3\,\frac{s}{s^2+9} + \frac{6}{s^2+9}$$

A inversa de $F_1(s)$ é $\cos(3\,t)$ e a inversa de $F_2(s)$ é $\sin(3\,t)$.

```
> f1 := invlaplace(F1, s, t):
  f2 := invlaplace(F2, s, t):
  f[1](t) , ` = L^-1 (F1(s)) ` = f1;
  f[2](t) , ` = L^-1 (F2(s)) ` = f2;
```

$$f_1(t),\ = L\text{^-}1\ (F1(s))\ = \cos(3\,t)$$

$$f_2(t),\ = L\text{^-}1\ (F2(s))\ = \sin(3\,t)$$

Portanto $\mathrm{f}(t)$ $=$ $3f_1(t) + 2f_2(t)$ $=$ $3\cos(3\,t) + 2\sin(3\,t)$.

```
> `f(t) = `, 3*f[1](t) + 2*f[2](t) = 3*f1 + 2*f2;
```

$$f(t) = ,\ 3f_1(t) + 2f_2(t) = 3\cos(3\,t) + 2\sin(3\,t)$$

Podemos verificar isto usando os procedimentos do Maple.

```
> `F(s) ` = (3*s + 6)/(s^2 + 9);
  `f(t) ` = invlaplace((3*s + 6)/(s^2 + 9), s, t);
```

$$F(s)\ = \frac{3\,s+6}{s^2+9}$$

$$f(t)\ = 3\cos(3\,t) + 2\sin(3\,t)$$

Transformada de Laplace de derivadas e integrais

Teorema (Derivada de f(t)) Seja $\mathrm{f}(t)$ e $f\,'(t)$ contínuas em $0 \le t$, e de ordem exponencial. Então, $\mathrm{L}(f\,'(t)) = s\,\mathrm{F}(s) - \mathrm{f}(0)$, onde $\mathrm{L}(f(t)) = \mathrm{F}(s)$.

Teorema (Integração de f(t)) Seja $\mathrm{f}(t)$ e $f\,'(t)$ contínuas para $0 \le t$, e de ordem exponencial . Então, $\mathrm{L}\left(\int \mathrm{f}(t)\,dt\right) = \frac{\mathrm{F}(s)}{s}$, onde $\mathrm{L}(\mathrm{f}(t)) = \mathrm{F}(s)$.

Carregue os procedimentos para transformada de Laplace. Faça isto com.

```
> with(inttrans):
```

Exemplo 1 Determine $L(\cos(t)^2)$.

```
> f:='f': F:='F': s:='s': t:='t': T:='T':
  f := t -> cos(t)^2:
  f1 := t -> subs(T=t,diff(f(T),T)):
  `f(t) ` = f(t);
  `f(0) ` = f(0);
  `f '(t) ` = f1(t);
```

$$f(t) = \cos(t)^2$$

$$f(0) = 1$$

$$f\,'(t) = -2\cos(t)\sin(t)$$

Use que $L(\sin(2\,t)) = \dfrac{2}{s^2+4}$.

```
> LDf := - 2/(s^2 + 4):
  LDf = `L(f '(t)) `;
  eqn :=  LDf = s*F(s) - f(0): eqn;
  sol := solve(eqn, F(s)):
  `Resolva para  F(s).`;
  `F(s) ` = sol;
  sol := simplify(sol):
  `F(s) ` = sol;
```

$$-\frac{2}{s^2+4} = L(f\,'(t))$$

$$-\frac{2}{s^2+4} = s\,F(s) - 1$$

Resolva para F(s).

$$F(s) = \frac{2+s^2}{s\,(s^2+4)}$$

$$F(s) = \frac{2+s^2}{s\,(s^2+4)}$$

Verifique isto usando as rotinas para Laplace

```
> `f(t) ` = cos(t)^2;
  `F(s) ` = laplace(cos(t)^2, t, s);
```

$$f(t) = \cos(t)^2$$

$$F(s) = \mathrm{laplace}(\cos(t)^2, t, s)$$

Surpresa, o Maple NÃO pode calcular!

Exemplo 2 Use o teorema acima apra determinar $\mathrm{L}(\mathrm{f}(t))$,onde

(a) $\mathrm{L}(t^2)$.

Como $f\,'(t) = 2\,t$ e $\mathrm{L}(2\,t) = \frac{2}{s^2}$.

```
> f:='f': F:='F': s:='s': t:='t':
  f := t -> t^2:
  f1 := t -> subs(T=t,diff(f(T),T)):
  `f(t) ` = f(t);
  `f '(t) ` = f1(t);
  LDf := laplace(f1(t), t, s):
  `L(f '(t)) ` = LDf;
  Lf := LDf/s:
  `F(s) = L(f '(t))/s ` = Lf;
```

$$f(t) = t^2$$

$$f\,'(t) = 2\,t$$

$$L(f\,'(t)) = \frac{2}{s^2}$$

$$F(s) = L(f\,'(t))/s = \frac{2}{s^3}$$

Verifique isto com as rotinas do Maple para Laplace.

```
> `f(t) ` = t^2;
  `F(s) ` = laplace(t^2, t, s);
```

$$f(t) = t^2$$

$$F(s) = \frac{2}{s^3}$$

(b) $\mathrm{L}(t^3)$.

Como $f\,'(t) = 3\,t^2$ e $\mathrm{L}(3\,t^2) = \frac{6}{s^3}$.

```
> f:='f': F:='F': s:='s': t:='t':
  f := t -> t^3:
  f1 := t -> subs(T=t,diff(f(T),T)):
  `f(t) ` = f(t);
  `f '(t) ` = f1(t);
  LDf := laplace(f1(t), t, s):
  `L(f '(t)) ` = LDf;
  Lf := LDf/s:
  `F(s) = L(f '(t))/s ` = Lf;
```

$$f(t) = t^3$$

$$f\,'(t) = 3\,t^2$$

$$L(f\,'(t)) = \frac{6}{s^3}$$

$$F(s) = L(f\,'(t))/s = \frac{6}{s^4}$$

Verifique isto com as rotinas do Maple.

```
> `f(t) ` = t^3;
  `F(s) ` = laplace(t^3, t, s);
```

$$f(t) = t^3$$

$$F(s) = \frac{6}{s^4}$$

Exemplo 3 Resolver o seguinte problema de valor inicial

$y''(t) + y(t) = 0$ com $y(0) = 2$ e $y'(0) = 3$.

```
> s:='s': t:='t': Y:='Y': Ys:='Ys':
  y0 := 2:
  y1 := 3:
  F := 0:
  `y''(t)  +  y(t)  =  0`;
  `y(0) ` = y0,`   y'(0) ` = y1;
  eqn :=  s^2*Y(s) - s*y0 - y1 + Y(s) = F: eqn;
  sol := simplify(solve(subs(s=S,eqn),Y(S))):
  Y := s -> subs(S=s,sol):
  `Y(s) ` = Y(s);
```

$$y''(t) + y(t) = 0$$

$$y(0) = 2, \quad y'(0) = 3$$

$$s^2 Y(s) - 2s - 3 + Y(s) = 0$$

$$Y(s) = \frac{2s+3}{s^2+1}$$

Usando que $L(\cos(t)) = \dfrac{s}{s^2+1}$ e $L(\sin(t)) = \dfrac{1}{s^2+1}$

vamos determinar a solução que é uma combinação linear de cos(t) e sin(t) .

```
> F1 := s/(s^2+1):
  F2 := 1/(s^2+1):
  `Y(s) ` = 2*F1 + 3*F2;
  f1 := invlaplace(F1, s , t):
  f2 := invlaplace(F2, s , t):
  `f(t) ` = 2*f1 + 3*f2;
```

$$Y(s) = 2\frac{s}{s^2+1} + \frac{3}{s^2+1}$$

$$f(t) = 2\cos(t) + 3\sin(t)$$

Podemos usar o Maple para determinar diretamente a inversa.

```
> `Y(s) ` = Y(s);
  `f(t) ` = invlaplace(Y(s), s , t);
```

$$Y(s) = \frac{2s+3}{s^2+1}$$

$$f(t) = 2\cos(t) + 3\sin(t)$$

Exemplo 4 Resolva o problema de valor inicial

$y''(t) + y'(t) - 2y(t) = 0$ com $y(0) = 1$ e $y'(0) = 4$.

```
> s:='s': S:='S': t:='t': Y:='Y':
  y0 := 1:
  y1 := 4:
  F := 0:
  `y''(t)  +  y'(t)  -  2y(t) =  0`;
  `y(0) ` = y0, `   y'(0) ` = y1;
  eqn :=  s^2*Y(s) - s*y0 - y1 + s*Y(s) - y0 - 2*Y(s) = F:
  eqn;
  sol := simplify(solve(subs(s=S,eqn),Y(S))):
  Y := s -> subs(S=s,sol):
  `Y(s) ` = Y(s);
  `Y(s) ` = convert(Y(s), parfrac, s);
```

$$y''(t) + y'(t) - 2y(t) = 0$$

$$y(0) = 1, \quad y'(0) = 4$$

$$s^2\,Y(s) - s - 5 + s\,Y(s) - 2\,Y(s) = 0$$

$$Y(s) = \frac{s+5}{s^2+s-2}$$

$$Y(s) = -\frac{1}{s+2} + \frac{2}{s-1}$$

Usando que $\mathrm{L}(\mathbf{e}^{(-2t)}) = \dfrac{1}{s+2}$ e $\mathrm{L}(\mathbf{e}^{t}) = \dfrac{1}{s-1}$

a solução é combinação linear .

```
> F1 := 1/(s+2):
  F2 := 1/(s-1):
  `Y(s) ` = - F1 + 2*F2;
  f1 := invlaplace(F1, s , t):
  f2 := invlaplace(F2, s , t):
  `f(t) ` = - f1 + 2*f2;
```

$$Y(s) = -\frac{1}{s+2} + \frac{2}{s-1}$$

$$f(t) = -e^{(-2\,t)} + 2\,e^{t}$$

Podemos usar a inversa de Laplace diretamente.

```
> `Y(s) ` = Y(s);
  `f(t) ` = invlaplace(Y(s), s , t);
```

$$Y(s) = \frac{s+5}{s^2+s-2}$$

$$f(t) = -e^{(-2\,t)} + 2\,e^{t}$$

Teoremas de Deslocamento

Teorema (deslocamento na variável s) Se $F(s)$ é a transformada de Laplace de $f(t)$,

então $L(e^{(a\,t)}\,f(t)) = F(s-a)$.

Teorema (deslocamento na variável t) Se $F(s)$ é a transformada de Laplace de $f(t)$,

e $0 \le a$, então $L(U_a(t)\,f(t-a)) = e^{(-a\,s)}\,F(s)$.

Carregue o pacote de transformações para trabalhar com o Maple.

```
> with(inttrans):
  laplace(t,t,s):
```

Exemplo 1 Calcule $L(t^n\,e^{(a\,t)})$.

Usando que $L(t^n) = \frac{n!}{s^{(n+1)}}$. e fazemos o deslocamento.

```
> a:='a': f:='f': F:='F': n:='n': s:='s': t:='t':
  f := t -> t^n:
  F := s ->  n!/s^(n+1):
  `formulas dadas:`;
  `f(t) ` = f(t);
```

```
`F(s) ` = F(s);
`deslocamento na variavel s para obter:`;
`f(t) ` = f(t)*exp(a*t);
`F(s) ` = F(s-a);
```

$$formulas\ dadas:$$

$$f(t) = t^n$$

$$F(s) = \frac{n!}{s^{(n+1)}}$$

$$deslocamento\ na\ variavel\ s\ para\ obter:$$

$$f(t) = t^n e^{(a\,t)}$$

$$F(s) = \frac{n!}{(-a+s)^{(n+1)}}$$

Podemos verificar isto com o Maple.

```
> assume(A,positive);
  assume(N,positive);
  `Por exemplo, comece com:`;
  `f(t) ` = t^n;
  L := laplace(t^N, t, s):
  `F(s) ` = subs(N='n',L);
  `Deslocamento na variável s para obter:`;
  `f(t) ` = t^n*exp(a*t);
  LS := laplace(t^N*exp(A*t), t, s):
  `F(s) ` = subs({A='a',N='n'},LS);;
```

$$Por\ exemplo,\ comece\ com:$$

$$f(t) = t^n$$

$$F(s) = \frac{\Gamma(n)\,n}{s^n\,s}$$

$$Deslocamento\ na\ variável\ s\ para\ obter:$$

$$f(t) = t^n e^{(a\,t)}$$

$$F(s) = \frac{\Gamma(n)\, n}{(-a+s)^n\,(-a+s)}$$

Exemplo 2 Resolva o PVI

$y''(t) + y(t) = U_\pi(t)$ com $y(0) = 0$ e $y'(0) = 0$.

```
> s:='s': S:='S': t:='t': Y:='Y': Y:='Y':
  y0 := 0:
  y1 := 0:
  F := laplace(Heaviside(t-Pi), t, s):
  `y''(t)  +  y(t)  =  UPi(t)`;
  `y(0) ` = y0,`   y'(0) ` = y1;
  eqn :=  s^2*Y(s) - s*y0 - y1 + Y(s) = F: eqn;
  sol := simplify(solve(subs(s=S,eqn),Y(S))):
  Y := s -> subs(S=s,sol):
  `Y(s) ` = Y(s);
  1/(s*(s^2+1)) = convert(1/(s*(s^2+1)), parfrac, s);
  `Y(s) ` = exp(-Pi*s)/s - exp(-Pi*s)*s/(s^2+1);
```

$$y''(t) + y(t) = UPi(t)$$

$$y(0) = 0, \quad y'(0) = 0$$

$$s^2\, Y(s) + Y(s) = \frac{e^{(-\pi s)}}{s}$$

$$Y(s) = \frac{e^{(-\pi s)}}{s\,(s^2+1)}$$

$$\frac{1}{s\,(s^2+1)} = \frac{1}{s} - \frac{s}{s^2+1}$$

$$Y(s) = \frac{e^{(-\pi s)}}{s} - \frac{e^{(-\pi s)}\, s}{s^2+1}$$

Use que $\mathrm{L}(U_\pi(t)) = \dfrac{e^{(-\pi s)}}{s}$ e $\mathrm{L}(U_\pi(t)\cos(t-\pi)) = \dfrac{e^{(-\pi s)}\, s}{s^2+1}$.

```
> F1 := exp(-Pi*s)/s:
  F2 := exp(-Pi*s)*s/(s^2+1):
```

```
f1 := invlaplace(F1, s , t):
f2 := invlaplace(F2, s , t):
`Y(s) ` = F1 - F2;
`f(t) ` = f1 - f2;
```

$$Y(s) = \frac{e^{(-\pi s)}}{s} - \frac{e^{(-\pi s)} s}{s^2+1}$$

$$f(t) = \text{Heaviside}(t-\pi) + \text{Heaviside}(t-\pi)\cos(t)$$

Podemos verificar isto usando o Maple.

```
> ODE  := diff(y(t),t$2)+y(t) = Heaviside(t-Pi):
  ICs := {y(0)=0, D(y)(0)=0}:
  `D. E.     ` = ODE;
  `I. C.'s   ` = ICs;
  dsolve(DE, y(t), method=laplace);
  dsolve({ODE} union ICs, y(t), method=laplace);
```

$$D.\ E. \quad = \left(\left(\frac{d}{dt}\left(\frac{d}{dt}y(t)\right)\right) + y(t) = \text{Heaviside}(t-\pi)\right)$$

$$I.\ C.'s \quad = \{D(y)(0) = 0, y(0) = 0\}$$

Error, (in dsolve) invalid arguments

$$y(t) = \text{Heaviside}(t-\pi) + \text{Heaviside}(t-\pi)\cos(t)$$

Invertendo a Transformada de Laplace

```
> with(inttrans):
  laplace(t,t,s):
```

Exemplo 1 Determine a transrformada inversa de $Y(s) = \frac{s^3 - 4s + 1}{s(s-1)^3}$.

```
> s:='s': Y:='Y':
  Y := s ->(s^3 - 4*s + 1)/(s*(s-1)^3):
  `Y(s) ` = Y(s);
```

$$Y(s) = \frac{s^3 - 4s + 1}{s(s-1)^3}$$

Vamos usar frações parciais para decompor a expressão de Y(s) em faotres mais simples .

```
> s:='s': Y:='Y':
  Y := s ->(s^3 - 4*s + 1)/(s*(s-1)^3):
  `Y(s) ` = Y(s);
  `Y(s) ` = convert(Y(s),parfrac,s);
```

$$Y(s) = \frac{s^3 - 4s + 1}{s(s-1)^3}$$

$$Y(s) = -\frac{1}{s} - \frac{2}{(s-1)^3} + \frac{1}{(s-1)^2} + \frac{2}{s-1}$$

Usamos a tabela de transformada de Laplace para procurar a inversa e obtemos a solução:

$$\mathrm{f}(t) = -1 - t^2\,\mathbf{e}^t + t\,\mathbf{e}^t + 2\,\mathbf{e}^t$$

Podemos usar o Maple para conferir.

```
> `F(s) ` = (s^3 - 4*s + 1)/(s*(s-1)^3);
  `f(t) ` = invlaplace((s^3 - 4*s + 1)/(s*(s-1)^3), s, t);
```

$$F(s) = \frac{s^3 - 4s + 1}{s(s-1)^3}$$

$$f(t) = -1 - t^2\,\mathbf{e}^t + t\,\mathbf{e}^t + 2\,\mathbf{e}^t$$

Exemplo 2 Determine a transformada inversa de $\mathrm{F}(s) = \dfrac{5s}{(s^2+4)(s^2+9)}$.

```
> s:='s': Y:='Y':
  Y := s -> 5*s/((s^2+4)*(s^2+9)):
  `Y(s) ` = Y(s);
```

$$Y(s) = 5\,\frac{s}{(s^2+4)(s^2+9)}$$

Podemos usar o Maple para converter em frações parciais.

```
> s:='s': Y:='Y':
  Y := s -> 5*s/((s^2+4)*(s^2+9)):
  `Y(s) ` = Y(s);
  `Y(s) ` = convert(Y(s),parfrac,s);
```

$$Y(s) = 5\frac{s}{(s^2+4)(s^2+9)}$$

$$Y(s) = \frac{s}{s^2+4} - \frac{s}{s^2+9}$$

Use a tabela de transformadad e Laplace para determinar a inversa:

$$y(t) = \cos(2\,t) - \cos(3\,t)$$.

Exemplo 3 Determine a transformada inversa de $Y(s) = \dfrac{s^3+3\,s^2-s+1}{s\,(s+1)^2\,(s^2+1)}$.

Em fraçoes parciais temos

```
> s:='s': Y:='Y':
  Y := s ->(s^3+3*s^2-s+1)/(s*(s+1)^2*(s^2+1)):
  `Y(s) ` = Y(s);
  `Y(s) ` = convert(Y(s),parfrac,s);
```

$$Y(s) = \frac{s^3+3\,s^2-s+1}{s\,(s+1)^2\,(s^2+1)}$$

$$Y(s) = \frac{1}{s} - \frac{2}{(s+1)^2} - \frac{2}{s+1} + \frac{s+1}{s^2+1}$$

>

Usando uma tabela de transformada de Laplace temos que a resposta é:

$$y(t) = 1 - 2\,t\,\mathbf{e}^{(-t)} - 2\,\mathbf{e}^{(-t)} + \cos(t) + \sin(t)$$.